
아날로그 입출력

아날로그 입출력



시계 방향 R_{12} 증가, R_{23} 감소

$$\text{입력 전압값} = \frac{\text{디지털 값}}{1023} \times 5[V]$$



다이얼 와이퍼

저항물질

아날로그 입출력

❖ 아날로그 입력 함수

void analogReference(type)			
기능	아날로그 핀에 입력되는 최대 전압 설정		
매개변수	type	DEFAULT	아두이노 보드 동작 전압(5V 또는 3.3V)
		INTERNAL	내부 기준전압 1.1V
		EXTERNAL	AREF 핀에 인가된 0~5V 전압
리턴 값	없음(void)		


int analogRead(pin_no)		
기능	지정된 아날로그 핀에 입력되는 아날로그 전압을 디지털로 변환하여 읽기	
매개변수	pin_no	아날로그 입력 핀 번호(0~5)
리턴 값	0~1023(10비트) 디지털 값	

아날로그 입출력

❖ 아날로그 입력 함수

long map(value, f1_val, f2_val, t1_val, t2_val)		
기능	값의 범위(최소, 최대)를 바꾸어주는 함수 (f1_val은 t1_val로, f2_val은 t2_val로 각각 변환되고, f1_val과 f2_val 상이의 값은 t1_val과 t2_val 상이의 값으로 변환된다.)	
매개변수	value	매핑 변환할 값
	f1_val	매핑 변환할 값의 최소값
	f2_val	매핑 변환할 값의 최대값
	t1_val	매핑 변환될 값의 최소값
	t2_val	매핑 변환될 값의 최대값
리턴 값	매핑 변환된 값	

❖ 회로구성



실습1: A/D 변환 결과 시리얼 모니터 창 표시

❖ ex01/app.ino

```
const int var_pin = A0; // 가변저항 연결핀
int analog_val; // 아날로그 값 저장 변수
```

```
void setup() {
  Serial.begin(115200);
}
```

```
void loop() {
  int digital_val;
  float ff;
```

$$\text{입력 전압값} = \frac{\text{디지털 값}}{1023} \times 5[V]$$

```
digital_val = analogRead(var_pin); // 아날로그 값 디지털 변환
ff = (float)digital_val / 1023. * 5.0; // 전압값 계산
```


```
Serial.print("Input Voltage(0~5V) = ");
Serial.println(ff); // 입력 전압값 출력(0~5V)
```

```
Serial.print("Digital Value(0~1023) = ");
Serial.println(digital_val); // 변환된 디지털 값 출력(0~1023)
Serial.println("-----");
```

```
delay(2000);
}
```

아날로그 입출력

❖ PWM 파형



아날로그 입출력

❖ 아날로그 출력 함수(PWM 출력함수)

void analogWrite(pin_no, value)		
기능	지정된 PWM 출력 핀에 0~255값 출력	
매개변수	pin_no	출력 핀 번호(3, 5, 6, 9, 10, 11) 3, 11(Timer2), 9, 10(Timer1) : 주기 약 490Hz 5, 6핀(Timer0) : 약 980Hz
	value	0~255 값(듀티 비 : $\text{value} / 255 * 100\%$) (0이면 항상 off, 255이면 항상 on)
리턴 값	없음(void)	

- PWM 출력 지원 핀
 - 3, 5, 6, 9, 10, 11

실습2: LED에 PWM 신호 출력하기

❖ ex02/app.ino

```
// LED에 PWM 신호 출력하기
const int led_pin = 9; // LED연결핀(PWM 출력핀)

void setup()
{
    pinMode(led_pin, OUTPUT); // PWM 출력 핀 출력 방향 설정
    digitalWrite(led_pin, 0); // LED OFF
}

void loop()
{
    int pwm_val;

    for (pwm_val = 0; pwm_val < 256; pwm_val += 5) {
        analogWrite(led_pin, pwm_val); // PWM 신호 출력
        delay(100);
    }

    digitalWrite(led_pin, 0); // LED OFF
    delay(2000);              // 2초 대기
}
```

❖ ex03/PWMLed.h

```
#pragma once

#include <Led.h>

class PWMLed: public Led {
protected:
    int value;
public:
    PWMLed(int pin, int value=0);
    void setValue(int value);
};
```

❖ ex03/PWMLed.cpp

```
#include "PWMLed.h"

PWMLed::PWMLed(int pin, int value): Led(pin), value(value) {
    analogWrite(pin, value);
}

void PWMLed::setValue(int value) {
    this->value = value;
    analogWrite(pin, value);
}
```

실습3: PWMLed 클래스로 PWM 신호 출력하기

❖ ex03/app.ino

```
#include "PWMLed.h"

PWMLed led(9);

void setup() {
}

void loop()
{
    int pwm_val;

    for (pwm_val = 0; pwm_val < 256; pwm_val += 5)
    {
        led.setValue(pwm_val); // PWM 신호 출력
        delay(100);
    }

    led.off();
    delay(2000);                // 2초 대기
}
```

실습4: LED에 PWM 신호 출력하기

❖ ex04/app.ino

```
#include <PWMLed.h>
const int var_pin = A0;    // 가변저항 연결핀
PWMLed led(9);

void setup() {
}

void loop()
{
    int  analog_val, pwm_val;

    analog_val = analogRead(var_pin);    // 가변저항 전압값 읽기
    pwm_val = map(analog_val, 0, 1023, 255, 0);    // 0~1023 -> 255~0으로 변환

    led.setValue(pwm_val);
}
```



실습5: Analog 클래스로 PWM 신호 출력하기

❖ ex05/Analog.h

```
#pragma once

#include <Arduino.h>

class Analog {
protected:
    int pin;
    int from_min;
    int from_max;
    int to_min;
    int to_max;

public:
    Analog(int pin, int from_min=0, int from_max=1023,
           int to_min=0, int to_max=255);
    void setRange(int from_min, int from_max, int to_min, int to_max);
    int read();
};
```

실습5: Analog 클래스로 PWM 신호 출력하기

❖ ex05/Analog.cpp

```
#include "Analog.h"

Analog::Analog(int pin, int from_min, int from_max, int to_min, int to_max):
    pin(pin), from_min(from_min), from_max(from_max),
    to_min(to_min), to_max(to_max) {
}

void Analog::setRange(int from_min, int from_max, int to_min, int to_max) {
    this->from_min = from_min;
    this->from_max = from_max;
    this->to_min = to_min;
    this->to_max = to_max;
}

int Analog::read() {
    int value = analogRead(pin);
    return map(value, from_min, from_max, to_min, to_max);
}
```

실습5: Analog 클래스로 PWM 신호 출력하기

❖ ex05/app.ino

```
#include "Analog.h"
#include <PWMLed.h>

Analog sensor(A0, 0, 1023, 255, 0);
PWMLed led(9);

void setup() {
}

void loop()
{
    int pwm_val = sensor.read();
    led.setValue(pwm_val);
}
```